Maple-assisted proof of formula for A251160

Robert Israel

11 January 2019

There are $4^2 = 16$ possible configurations for a 1×2 sub-array. Consider the 16×16 transition matrix T such that $T_{ij} = 1$ if the bottom row of a 2×2 sub-array could be in configuration i while the top row is in configuration j. The following Maple code computes it. I'm encoding a configuration $[b_1, b_2]$ as $1 + b_1 + 4\, b_2$. The $+1$ is needed because matrix indices start at 1 rather than 0.

```
> q:= proc(a,b) local a1, a2, b1, b2;
    a1:= (a-1) mod 4; a2:= (a-1-a1)/4;
    b1:= (b-1) mod 4; b2:= (b-1-b1)/4;
    if max(b1,a2) > abs(b2-a1) then 0 else 1 fi
  end proc:
> T:= Matrix(16,16,q):
```

Thus $a(n) = u\, T^n\, v$ where u and v are row and column vectors of all 1's.

```
> u:= Vector[row](16,1): v:= Vector(16,1):
```

To check, here are the first few entries of our sequence.

```
> [seq(u . T^n . v, n = 1 .. 20)];
```

[96, 552, 2658, 12001, 55131, 257417, 1201970, 5597648, 26056421, 121329295, 565030902, 2631278472, 12253239453, 57060424477, 265717806149, 1237389994220, 5762253389058, 26833543568447, 124957900541999, 581901431575301] **(1)**

Now here is the minimal polynomial P of T, as computed by Maple.

```
> LinearAlgebra:-MinimalPolynomial(T, t);
```

$$t^{14} - 7\,t^{13} + 18\,t^{12} - 43\,t^{11} + 59\,t^{10} - 70\,t^9 + 62\,t^8 - 33\,t^7 + 14\,t^6 + 3\,t^5 - 6\,t^4 + t^3 \qquad \textbf{(2)}$$

Thus for any $n \geq 0$ we will have

$$0 = u\,P(T)\,T^n\,v = \sum_{i=3}^{14} p_i\,a(i+n) = a(n+3) - 6\,a(n+4) + 3\,a(n+5) + 14\,a(n+6) - 33\,a(n+7) + 62\,a(n+8) - 70\,a(n+9) + 59\,a(n+10) - 43\,a(n+11) + 18\,a(n+12) - 7\,a(n+13) + a(n+14)$$

where p_i is the coefficient of t^i in $P(t)$. This is equivalent to the empirical formula.